graph twoway line — Two-way line plots

Description

`line` draws line plots.

`line` is a command and a *plottype* as defined in [G-2] **graph twoway**. Thus the syntax for `line` is

```
. graph twoway line ...
. twoway line ...
. line ...
```

Being a plottype, `line` may be combined with other plottypes in the `twoway` family (see [G-2] **graph twoway**), as in

```
. twoway (line ...) (scatter ...) (lfit ...) ...
```

which can equivalently be written as

```
. line ... || scatter ... || lfit ... || ...
```

Quick start

A line plot of `y1` versus `x`

```
twoway line y1 x
```

Same as above, but sort on values of `x`

```
twoway line y1 x, sort
```

A line plot of `y1`, `y2`, and `y3` each against sorted values of `x`

```
twoway line y1 y2 y3 x, sort
```

Same as above, but specify a different pattern for each line

```
twoway line y1 y2 y3 x, sort lpattern(dash solid dot)
```

Plot lines in a separate graph area for each level of `catvar`

```
twoway line y1 y2 y3 x, sort by(catvar)
```

Add "My Title" as an overall graph title

```
twoway line y1 y2 y3 x, sort by(catvar, title("My Title"))
```

Same as above, but add "My Title" as the title of each subgraph

```
twoway line y1 y2 y3 x, sort by(catvar) title("My Title")
```

Menu

Graphics > Two-way graph (scatter, line, etc.)

Syntax

[twoway] line *varlist* [*if*] [*in*] [, *options*]

where *varlist* is

y_1 [y_2 [...]] x

options	Description
connect_options	change look of lines or connecting method
colorvar_options	change color of lines based on values of a variable
axis_choice_options	associate plot with alternative axis
twoway_options	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

connect_options discusses options for one y versus one x; see *connect_options* in [G-2] **graph twoway scatter** when plotting multiple ys against one x.

Options

connect_options specify how the points forming the line are connected and the look of the lines, including pattern, width, and color; see [G-3] ***connect_options***.

[G-3] ***connect_options*** discusses options for one *y* versus one *x*, see *connect_options* in [G-2] **graph twoway scatter** when plotting multiple *y*s against one *x*.

colorvar_options specify that the color of the lines be determined by the levels of the numeric variable *colorvar*; see [G-3] ***colorvar_options***. *colorvar_options* are not allowed when plotting multiple ys against one x.

axis_choice_options associate the plot with a particular y or x axis on the graph; see [G-3] ***axis_choice_options***.

twoway_options are a set of common options supported by all twoway graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over by() groups, and change some advanced settings. See [G-3] ***twoway_options***.

Remarks and examples

Remarks are presented under the following headings:

One-way equivalency of line and scatter

`line` is similar to `scatter`, the differences being that by default the marker symbols are not displayed and the points are connected:

Default `msymbol()` option: `msymbol(none ...)`

Default `connect()` option: `connect(l ...)`

Thus you get the same results typing

```
. line yvar xvar
```

as typing

```
. scatter yvar xvar, msymbol(none) connect(l)
```

You can use `scatter` in place of `line`, but you may not use `line` in place of `scatter`. Typing

```
. line yvar xvar, msymbol(O) connect(none)
```

will not achieve the same results as

```
. scatter yvar xvar
```

because `line`, while it allows you to specify the *marker_option* `msymbol()`, ignores its setting.

Typical use

`line` draws line charts:

```
. use https://www.stata-press.com/data/r19/uslifeexp
(US life expectancy, 1900-1999)
. line le year
```



Line charts work well with time-series data. With other datasets, lines are often used to show predicted values and confidence intervals:

```
. use https://www.stata-press.com/data/r19/auto, clear
(1978 automobile data)
. quietly regress mpg weight
. predict hat
(option xb assumed; fitted values)
. predict stdf, stdf
. generate lo = hat - 1.96*stdf
. generate hi = hat + 1.96*stdf
. scatter mpg weight || line hat lo hi weight, pstyle(p2 p3 p3) sort
```



Do not forget to include the `sort` option when the data are not in the order of the x variable, as they are not above. We also included `pstyle(p2 p3 p3)` to give the lower and upper confidence limit lines the same look; see *Appendix: Styles and composite styles* under *Remarks and examples* in [G-2] **graph twoway scatter**.

Because `line` is `scatter`, we can use any of the options allowed by `scatter`. Below we return to the US life expectancy data and graph black and white male life expectancies, along with the difference, specifying many options to create an informative and visually pleasing graph:

```
. use https://www.stata-press.com/data/r19/uslifeexp, clear
(US life expectancy, 1900-1999)
. generate diff = le_wm - le_bm
. label var diff "Difference"
.    line le_wm year, yaxis(1 2) xaxis(1 2)
  || line le_bm year
  || line diff  year
  || lfit diff  year
  ||,
        ytitle("",          axis(2))
        xtitle("",          axis(2))
        xlabel(1918,        axis(2))
        ylabel(0(5)20,      axis(2) grid)
        ylabel(0 20(10)80               )
        ytitle("Life expectancy at birth (years)")
        title("White and black life expectancy")
        subtitle("USA, 1900 to 1999")
        note("Source: National Vital Statistics, Vol 50, No. 6"
             "(1918 dip caused by 1918 influenza pandemic)")
```



See [G-2] **graph twoway scatter**.

Advanced use

The above graph would look better if we shortened the descriptive text used in the keys. Below we add

```
legend(label(1 "White male") label(2 "Black male"))
```

to our previous command:

```
. line le_wm year, yaxis(1 2) xaxis(1 2)
  || line le_bm year
  || line diff  year
  || lfit diff  year
  ||,
        ytitle("",          axis(2))
        xtitle("",          axis(2))
        xlabel(1918,        axis(2))
        ylabel(0(5)20,      axis(2) grid)
        ylabel(0 20(10)80               )
        ytitle("Life expectancy at birth (years)")
        title("White and black life expectancy")
        subtitle("USA, 1900 to 1999")
        note("Source: National Vital Statistics, Vol 50, No. 6"
             "(1918 dip caused by 1918 influenza pandemic)")
        legend(label(1 "White male") label(2 "Black male"))
```



Cautions

Be sure that the data are in the order of the x variable, or specify `line`'s `sort` option. If you do neither, you will get something that looks like the scribblings of a child:

```
. use https://www.stata-press.com/data/r19/auto, clear
(1978 automobile data)
. line mpg weight
```



Reference

Christodoulou, D. 2017. Heuristic criteria for selecting an optimal aspect ratio in a two-variable line plot. *Stata Journal* 17: 279–313.

Also see

[G-2] **graph twoway scatter** — Two-way scatterplots

[G-2] **graph twoway fpfit** — Two-way fractional-polynomial prediction plots

[G-2] **graph twoway lfit** — Two-way linear prediction plots

[G-2] **graph twoway mband** — Two-way median-band plots

[G-2] **graph twoway mspline** — Two-way median-spline plots

[G-2] **graph twoway qfit** — Two-way quadratic prediction plots